ВИЗУАЛИЗАЦИЯ КОЛЕБАТЕЛЬНОГО ДВИЖЕНИЯ СВЯЗАННЫХ МЕХАНИЧЕСКИХ СИСТЕМ В СРЕДЕ MATHEMATICA

Андреев В.В.
Профессор кафедры теоретической физики, д.ф.-м.н., профессор, УО «Гомельский государственный университет имени Франциска Скорины», Республика Беларусь, г. Гомель

Максименко Н.В.
Профессор кафедры теоретической физики, д.ф.-м.н., профессор, УО «Гомельский государственный университет имени Франциска Скорины», Республика Беларусь, г. Гомель

Дерюжкова О.М.
Доцент кафедры теоретической физики, к.ф.-м.н., доцент, УО «Гомельский государственный университет имени Франциска Скорины», Беларусь, г. Гомель

Аннотация. В работе продемонстрированы возможности системы Wolfram Mathematica для исследования и визуализации колебаний системы связанных маятников.

Ключевые слова: Wolfram Mathematica, колебательные процессы, связанные математические маятники.

Рассмотрим связанную механическую систему, движущуюся в одной плоскости и имеющую n степеней свободы. Система из n связанных математических маятников состоит из n невесомых стержней и n материальных точек одинаковой или разной длины и массы, причем каждый последующий маятник подвешен к грузу предыдущего. Маятники, как правило, вращаются вокруг горизонтальной оси так, что все точки подвесов лежат в одной плоскости и составляют с вертикалью некоторые углы. Поэтому, общепринято для получения уравнения движения такой системы вводить углы отклонения маятников от положения равновесия [1]. В данной работе продемонстрирован пример использования декартовых координат для описания уравнений движений (Рис. 1).



Рис. 1. Фрагмент системы связанных маятников

В отличие от случая с использованием угловых переменных, система дифференциальных уравнений в декартовых координатах дополняется алгебраическими уравнениями, которые возникают из условия того, что стержни каждого из маятников не изменяют своей длины. На рисунке 1 представлен фрагмент системы связанных маятников, который позволяет составить систему уравнений для i - того участка:

$$\left(\begin{array}{c} m_i \ddot{x}_i(t) = \dfrac{c_i(t)}{\ell_i}(x_i(t) - x_{i-1}(t)) - \dfrac{c_{i+1}(t)}{\ell_{i+1}}(x_{i+1}(t) - x_i(t)), \\ m_y \ddot{y}_i(t) = \dfrac{c_i(t)}{\ell_i}(y_i(t) - y_{i-1}(t)) - \dfrac{c_{i+1}(t)}{\ell_{i+1}}(y_{i+1}(t) - y_i(t)) - m_i g, \\ (x_i(t) - x_{i-1}(t))^2 + (y_i(t) - y_{i-1}(t))^2 = \ell_i^2 \end{array}\right), i = 1, 2, \ldots n \qquad (1)$$

Решение систем уравнений вида (1) является не простой задачей, однако современные системы знаний, такие как Wolfram Mathematica, позволяют провести численные расчеты и визуализировать полученные результаты в максимально наглядном виде [2].

Моделирование движения системы связанных математических маятников разбивается на несколько этапов. На первом этапе создается функция, позволяющая сформировать уравнения движения, начальные условия и список функций, которые должны быть вычислены. На Рис. 2 представлен программный модуль NPendulumDAESystem в среде Wolfram Mathematica, который генерирует информацию, необходимую для численного решения системы уравнений колебаний n-маятников с различными длинами и массами материальных точек.

```
NPendulumDAESystem[np_, length0_, mass0_, ϕ0_, g_, t_] :=
  Module[{eqns, vars, ics, length1 = length0, length, mass1 = mass0, mass, ϕ1 = ϕ0, ϕ, tabx, tabxn},
   x[0][t] = 0; y[0][t] = 0; c[np + 1][t] = 0; length[i_] := length1[[i]] /; i ≤ np;
   mass[i_] := mass1[[i]] /; i ≤ np; ϕ[i_] := ϕ1[[i]] /; i ≤ np;
   tabx = Flatten[Table[{x[i][0], y[i][0]}, {i, np}]];
   tabxn =
    tabx //. Flatten[Table[{x[i][0] → x[i - 1][0] + length[i] Sin[ϕ[i]], y[i][0] → y[i - 1][0] - length[i] Cos[ϕ[i]]},
       {i, np}]] //. {x[0][0] → 0, y[0][0] → 0};
   eqns = Flatten[Table[{
        mass[i] × x[i]''[t] == 1/length[i] c[i][t] (x[i][t] - x[i - 1][t]) -
          1/length[i+1] c[i + 1][t] (x[i + 1][t] - x[i][t]),
        mass[i] × y[i]''[t] == 1/length[i] c[i][t] (y[i][t] - y[i - 1][t]) -
          1/length[i+1] * c[i + 1][t] (y[i + 1][t] - y[i][t]) - mass[i] g,
        (x[i][t] - x[i - 1][t])^2 + (y[i][t] - y[i - 1][t])^2 == length[i]^2},
      {i, np}]];
   vars = Flatten[Table[{x[i], y[i], c[i]}, {i, np}]];
   ics = Flatten[{Table[tabx[[i]] == tabxn[[i]], {i, Length[tabx]}], Table[y[i]'[0] == 0, {i, np}]}];
   {eqns, vars, ics}];
```

Рис. 2. Программный модуль для создания системы уравнений движения связанных маятников

На втором этапе с помощью NPendulumDAESystem формируются данные для необходимого количества маятников, с определением конкретных числовых значений для начальных условий и для величины ускорения свободного падения g. Задается также временной интервал $t_{\max}$, на котором будут найдены решения. Далее эта информация используется в операторе NDSolve для получения численных решений. С целью контроля точности определена функция EnergyConservation, которая рассчитывает полную энергию данной системы в различные моменты времени.

На Рис. 3 представлен пример решения уравнений движения системы трех связанных математических маятников.

```
(* Начальные условия *)
numPen = 3; (* число маятников*)
lengthPen = {1.5, 1, 0.75} (* Длинны маятников*);
mass = {1, 0.5, 0.2}; (* массы маятников *)
φ = {60 Degree, 35 Degree, -10 Degree} ; (* углы отклонения от равновесия*)
g = 9.81; (* ускорение свободного падения *)
(* Генерация уравнений, начальных условий для оператора NDSolve *)
{eqns, vars, ics} = NPendulumDAESystem[numPen, lengthPen, mass, φ, g, t];
(* Граница времени для решений*)
tmax = 13.8;
(* Решение системы уравнений *)
sol =
  Quiet@NDSolve[{eqns, ics}, vars, {t, 0, tmax},
    Method → {"IndexReduction" → {Automatic, "IndexGoal" → 0}, "EquationSimplification" → "Residual"}];
(* Энергия, как функция времени *)
EnergyConservation[LengthPen_, numPen_, mass_, g_, t_] := Module[{energy, sol1},

   sol1 = sol[[1]];

   energy[t] = Sum[mass[[i]] * g * (Total[LengthPen] - y[i][t]), {i, 1, numPen}] -
     Sum[mass[[i]]/2 (x[i]'[t]^2 + y[i]'[t]^2), {i, 1, numPen}] //. sol1];
```

Рис. 3. Пример работы для системы из трех маятников

Третий этап состоит в визуализации колебаний системы. Wolfram Mathematica обладает богатым арсеналом отображения различной графической информации с использованием операторов типа Plot, ParametricPlot. Также имеются возможности анимации изображения с помощью операторов Animate, Manipulate и др.

На рис. 4 представлена временная зависимость координат материальных точек и сил натяжения для системы трех маятников. В нижней части рисунка отображено изменение полной энергии $E(t)$ в зависимости от времени, что позволяет судить о стабильности вычислительного процесса и его точности.



Рис. 4. Графическое отображение решения уравнений движения системы трех маятников с помощью оператора Plot и ListPlot

Рис. 5 демонстрирует скриншот анимации движения системы трех маятников, которая была получена с помощью оператора Manipulate с визуализацией траекторий движения материальных точек. Из рисунка видно, что второй и третий маятники совершают хаотические движения. Этот пример визуально демонстрирует уже хорошо известный факт: траектории $2,3,\ldots n$ материальных точек в системе связанных маятников чрезвычайно зависимы от начальных условий. Созданный программный модуль позволяет исследовать такую зависимость в наглядном виде. Так можно установить, что хаотическое поведение маятников часто приводит к тому, что численные расчеты можно проделать только на определенном временном интервале. Вне данного интервала хаотичность движения способствует стремительному росту ошибки вычислений.



Рис. 5. Графическое отображение решения уравнений движения системы из трех маятников с помощью Manipulate

Использование возможностей среды Wolfram Mathematica для моделирования и визуализации колебательного движения систем с различным количеством связанных математических маятников и при любых начальных и граничных условиях позволяет в первую очередь, значительно снизить трудоёмкость численных расчетов. Наличие графики и анимации дает визуальную картину, которая, благодаря своей наглядности, дополняет численные результаты, делая абстрактную механическую модель более реальной.

Список литературы:

1. Горелик Г.С. Колебания и волны. Введение в акустику, радиофизику и оптику. 3-е изд. М.: Физматлит, 2007. 656 с.
2. Wolfram S. The Mathematica book. Addison-Wesley, 1999. 359 p.